

MASSACHUSETTS INSTITUTE OF TECHNOLOGY
 Department of Electrical Engineering and Computer Science
 6.001—Structure and Interpretation of Computer Programs
 Fall 2007

Recitation 4 — 9/14/2007
Orders of Growth

Definitions

Theta (Θ) notation:

$$f(n) = \Theta(g(n)) \rightarrow k_1 \cdot g(n) \leq f(n) \leq k_2 \cdot g(n), \text{ for } n > n_0$$

Big-O notation:

$$f(n) = O(g(n)) \rightarrow f(n) \leq k \cdot g(n), \text{ for } n > n_0$$

Adversarial approach: For you to show that $f(n) = \Theta(g(n))$, you pick k_1 , k_2 , and n_0 , then I (the adversary) try to pick an n which doesn't satisfy $k_1 \cdot g(n) \leq f(n) \leq k_2 \cdot g(n)$.

Implications

Ignore constants. Ignore lower order terms. For a sum, take the larger term. For a product, multiply the two terms. Orders of growth are concerned with how the effort scales up as the size of the problem increases, rather than an exact measure of the cost.

Typical Orders of Growth

- $\Theta(1)$ - Constant growth. Simple, non-looping, non-decomposable operations have constant growth.
- $\Theta(\log n)$ - Logarithmic growth. At each iteration, the problem size is scaled down by a constant amount: (`call-again (/ n c)`).
- $\Theta(n)$ - Linear growth. At each iteration, the problem size is decremented by a constant amount: (`call-again (- n c)`).
- $\Theta(n \log n)$ - Nifty growth. Nice recursive solution to normally $\Theta(n^2)$ problem.
- $\Theta(n^2)$ - Quadratic growth. Computing correspondence between a set of n things, or doing something of cost n to all n things both result in quadratic growth.
- $\Theta(2^n)$ - Exponential growth. Really bad. Searching all possibilities usually results in exponential growth.

What's n ?

Order of growth is *always* in terms of the size of the problem. Without stating what the problem is, and what is considered primitive (what is being counted as a “unit of work” or “unit of space”), the order of growth doesn't have any meaning.

Problems

1. Give order notation for the following:

(a) $5n^2 + n$
 $\Theta(n^2)$

(b) $\sqrt{n} + n$
 $\Theta(n)$

(c) $3^n n^2$
 $\Theta(3^n n^2)$

(d) $\log_{57} n + \frac{1}{n^2}$
 $\Theta(\log n)$

2. What is the order of growth of the following procedure to calculate b^n by repeated multiplication?

```
(define (expt b n)
  (if (= n 0)
      1
      (* b (expt b (- n 1)))))
```

Running time? $\Theta(n)$ Space? $\Theta(n)$

3. Rewrite the previous definition so that it yields an iterative process.

```
(define (expt-iter b n)
  (define (helper p i)
    (if (= i 0) p
        (helper (* b p) (- i 1))))
  (helper 1 n))
```

Running time? $\Theta(n)$ Space? $\Theta(1)$

4. Now write a version of `expt` that takes less than $\Theta(n)$ time.

```
(define (fast-expt b n)
  (define (square x) (* x x))
  (define (help n)
    (cond ((= n 0) 1)
          ((even? n)
           (square (help (quotient n 2))))
          (else
           (* b (help (- n 1)))))
    (help n))
```

Running time: $\Theta(\log n)$ Space: $\Theta(\log n)$

5. One last `expt` version: This time, both time and space must be less than $\Theta(n)$.

```
(define (faster-expt b n)
  (define (square x) (* x x))
  (define (help r b n)
    (cond ((= n 0) r)
          ((even? n)
           (help r (square b) (quotient n 2)))
          (else
           (help (* b r) b (- n 1)))))
  (help 1 b n))
```

Running time: $\Theta(\log n)$ Space: $\Theta(1)$

6. Consider the recursive definition of factorial we've seen before:

```
(define (fact n)
  (if (= n 0)
      1
      (* n (fact (- n 1)))))
```

Running time? $\Theta(n)$ Space? $\Theta(n)$

Now look at the recursive version of `find-e` from last time:

```
(define (find-e n)
  (if (= n 0)
      1.
      (+ (/ (fact n)) (find-e (- n 1)))))
```

What is the resulting order of growth of `find-e`?

Running time? $\Theta(n^2)$ Space? $\Theta(n)$

7. Assume you have a procedure (`divisible? n x`) which returns `#t` if `n` is divisible by `x`. It runs in $O(n)$ time and $O(1)$ space. Write a procedure `prime?` which takes a number and returns `#t` if it's prime and `#f` otherwise. You'll want to use a helper procedure.

```
(define (prime? p)
  (define (helper n)
    (if (> n (sqrt p))
        #t
        (if (divisible? p n)
            #f
            (helper (+ n 1)))))
  (helper 2))
; iterative process
; assuming sqrt takes  $O(1)$  time,  $\text{sqrt}(n) * n$ 
```

Running time? $O(n\sqrt{n})$ Space? $O(1)$

8. Write a procedure that will multiply two positive integers together, but the only arithmetic operation allowed is addition (ie multiplication through repeated addition). In addition, your procedure should be iterative, not recursive. What is its order of growth?

(`slow-mul 3 4`) \rightarrow 12

```
(define (mul-helper a b total)
  (if (= a 0)
      total
      (mul-helper (- a 1) b (+ total b)))) ; or (+ a -1) if picky
(define (slow-mul a b)
  (mul-helper a b 0))
```

Running time? $\Theta(n)$ Space? $\Theta(1)$